TEMPLATE = app
TARGET = jcws
CONFIG += qt \
	warn_on \
	debug
QT = core \
	gui \
	xml
unix {
	isEmpty(PREFIX):PREFIX = /usr
	BINDIR = $$PREFIX/bin
	DATADIR = $$PREFIX/share/games/jcws
	CRWDDIR = $$DATADIR/crwd
	LOCALEDIR = $$DATADIR/locale
	DEFINES += LOCALEDIR=\\\"$$LOCALEDIR\\\"
	DEFINES += CRWDDIR=\\\"$$CRWDDIR\\\"
	INSTALLS += target \
		desktop \
		appicon \
		menu \
		crosswords \
		locale \
		mime
	target.path = $$BINDIR
	desktop.path = $$PREFIX/share/applications
	desktop.files += jcws.desktop
	appicon.path = $$PREFIX/share/pixmaps
	appicon.files += pics/jcws.png
	menu.path = $$PREFIX/share/menu
	menu.files += jcws
	crosswords.path = $$CRWDDIR
	crosswords.files += crws/*.jcw
	locale.path = $$LOCALEDIR
	locale.files += locale/*.qm
	mime.path = $$PREFIX/share/mime/packages
	mime.files += mime/packages/jcws.xml
}
SOURCES = src/main.cpp \
	src/mainwnd.cpp \
	src/abstractcrossword.cpp \
	src/rectcrossword.cpp \
	src/classiccw.cpp \
	src/crosswordinfodialog.cpp \
	src/fieldcheckerthread.cpp \
	src/newcwdialog.cpp \
	src/colorcw.cpp \
	src/palette.cpp \
	src/colorselector.cpp
HEADERS = src/mainwnd.h \
	src/abstractcrossword.h \
	src/rectcrossword.h \
	src/classiccw.h \
	src/crosswordinfodialog.h \
	src/fieldcheckerthread.h \
	src/newcwdialog.h \
	src/cwtype.h \
	src/colorcw.h \
	src/palette.h \
	src/colorselector.h
RESOURCES = src/jcws.qrc
TRANSLATIONS = locale/jcws_r_RU.ts
DESTDIR = bin
OBJECTS_DIR = build
MOC_DIR = build
UI_DIR = build
